

Données sur le trafic des compagnies aériennes reportées par les principaux aéroports européens

Projet DANA - Web des données, web sémantique
Master 1 - ALMA - 2022-2023

Quentin GOMES DOS REIS
Alan PROVOST
Ahmat IZZEDINE ISSA

Sommaire

- Étape n°1 : Choix et sémantisation d'un dataset
 - Présentation du jeu de données
 - Attributs du dataset
 - Attributs du dataset - Types de mesures
 - Exemple de données
 - Sémantisation des données
 - Choix effectué
 - Méthode de sémantisation
 - Structure
 - Exemple de requêtes
- Étape n°2 : Liaison de notre dataset à d'autres datasets
 - Liaisons possibles
 - La principale difficulté
 - Solution employée
 - Liaisons réalisées et les requêtes
- Étape n°4 : Liaison de notre dataset au cloud LinkedData
- Étape n°5 : Description avec le vocabulaire VOID



Choix et sémantisation d'un dataset

Étape n°1

Présentation du jeu de données choisi



Présentation du jeu de données



eurostat

- **Intitulé** : Données sur le trafic des compagnies aériennes par principaux aéroports
- **Éditeur** : Eurostat
- **Code de données en ligne** : AVIA_TF_ALA
- **Licence** : CC BY 4.0
- **Date de mise à jour** : 25/11/2022 11:00 (utilisée: 24/10/22)
- **Fréquence de mise à jour** : Très régulière
- **Format** : Fichier CSV
- **Taille** : ~13 946 000 lignes / 9 colonnes / 1,3 Go

Lien vers le dataset:

- https://ec.europa.eu/eurostat/databrowser/view/AVIA_TF_ALA/default/table

Attributs du dataset

Colonne	Description	Valeurs possibles
freq	Fréquence d'échantillonnage	Années [A], Trimestriel [Q], Mensuel [M] (de 1993 à aujourd'hui)
unit	L'unité de mesure	Passager [PAS], Vol [FLIGHT], Tonne [T], Places assises et couchées [SEAT]
tra_meas	La métrique mesurée	Passagers à bord [PAS_BRD], Sièges passagers offerts [ST_PAS], Fret et courrier à bord [FRM_BRD], etc...
tra_cov	Couverture de transport	Transport total [TOTAL], national [NAT], international [INTL], intraUE, extraUE, etc...
airline	Origine de la compagnie aérienne	Toutes les compagnies [TOTAL], Compagnies licenciées dans des pays de l'UE [LIC_UE], Compagnies licenciées dans des pays non-UE [LIC_NUE], Compagnies dont les pays de licence est inconnu [UNK]
rep_airp	Aéroport déclarant	ANTWERPEN/DEURNE airport [BE_EBAW], PARDUBICE airport [CZ_LKPD], PARIS-ORLY airport [FR_LFPO], etc...
TIME_PERIOD	Année, mois ou trimestre	1993 [A], 2010-09 [M], 2022-Q2 [Q]
OBS_VALUE	valeur observée	Valeur numérique

Attributs du dataset - Types de mesures

- ☐ **PAS_BRD** : Passagers à bord
- ☐ **PAS_CRD** : Passagers transportés
- ☐ **ST_PAS** : Sièges passagers offerts
- ☐ **FRM_BRD** : Fret et courrier à bord
- ☐ **FRM_LD_NLD** : Fret et courrier chargés et déchargés
- ☐ **CAF** : Vols commerciaux (passagers, fret et courrier)
- ☐ **CAF_PAS** : Vols commerciaux de passagers
- ☐ **CAF_FRM** : Vols commerciaux fret et courrier

Exemple de données

FREQ	UNIT	TRA_MEAS	TRA_COV	AIRLINE	REP_AIRP	TIME_PERIOD AS NUMBER(INTEGER)	OBS_VALUE
A	T	FRM_BRD	NAT	LIC_EU	FR_LFPG	2010	44935.1
A	T	FRM_BRD	NAT	LIC_NEU	FR_LFPG	2010	401.1
A	T	FRM_BRD	NAT	TOTAL	FR_LFPG	2010	45336.2

Cet exemple décrit des données sur la période de l'année 2010 et sur l'aéroport Charles de Gaulle.

Les données correspondent à la masse de fret et/ou courrier à bord pour les compagnies européennes ou non ainsi que le total pour cette année.


Sémantisation des données



Choix effectué

En l'état, notre jeu de données est trop volumineux, nous ne pouvons pas traiter ce dernier, pour que cela soit possible, nous allons choisir quelles données nous allons garder.

Données	Nombre d'éléments	Répartition
Annuelles	1 441 267	~ 10,3%
Mensuelles	9 118 701	~ 65,4%
Trimestrielles	3 386 463	~ 24,3%
Total	13 946 431	




Afin de pouvoir traiter ce dernier, nous allons choisir de ne garder uniquement les données annuelles même si elles ne représentent que 10,3% de notre dataset

Méthode de sémantisation

Afin d'effectuer le processus de sémantisation de notre jeu de données, nous avons débuté en utilisant OpenRefine.

Après quelques problèmes liés à nos ressources matérielles, nous sommes passées sur TARQL et nous avons utilisé SPARQL pour pouvoir transcrire notre fichier CSV en un fichier Turtle correct.



Méthode de sémantisation

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX ex: <http://example.org/>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

CONSTRUCT {
  # Each measures
  ?measure
    ex:declarativeAirport      ?airport;
    dbo:year                    ?year_int;
    rdf:value                   ?value;
    ex:unit                     ?unit;
    ex:measuredMetric           ?tra_meas;
    ex:coverageTransport        ?tra_cov;
    ex:airline                  ?airline;
    rdf:type                    ex:AirportMetricValue.

  # Each rep_air attributes is transformed into an airport individual
  ?airport
    dbo:iso31661Code            ?countryCodeA3; # ISO31661-1 ALPHA 3
    dbo:iso31661Code            ?countryCodeA2; # ISO31661-1 ALPHA 2
    dbo:country                 ?countryName;   # English country name
    dbo:icaoLocationIdentifier    ?icaoID;        # ICAO - 4 LETTERS ID
    rdf:type                    dbo:Airport.
}
FROM <file:../raw_data/avia_tf_ala_linear.csv#encoding=utf-8;header=present;delimiter=comma>
WHERE {
  BIND (xsd:integer(?TIME_PERIOD) AS ?year_int). # Convert String to integer

  # Useful to have both float and int values in our resulting file
  BIND (IF (?unit = "T", xsd:float(?OBS_VALUE), xsd:integer(?OBS_VALUE)) AS ?value).

  # To get countryCode and icaoID from the rep_air attribute
  BIND (SUBSTR(STR(?rep_air), 1, 2) AS ?countryCode). # Raw countryCode from Dataset
  BIND (SUBSTR(STR(?rep_air), 4, 4) AS ?icaoID).
  BIND (IF (?countryCode = "EL", "GL", ?countryCode) AS ?countryCodeA2)


  # CONVERSION FROM ALPHA 2 TO ALPHA 3
  BIND (IF (?countryCode = "AT", "AUT",
    IF (?countryCode = "BE", "BEL",
      # Shortened for presentation purposes but we are doing this for each country
      IF (?countryCode = "UK", "GBR", "TUR"))))))))))))))))))))))))))))
    AS ?countryCodeA3).

  # CONVERSION FROM ALPHA 2 TO Country Name
  BIND (IF (?countryCode = "AT", "Austria",
    IF (?countryCode = "BE", "Belgium",
      # Shortened for presentation purposes but we are doing this for each country
      IF (?countryCode = "UK", "United Kingdom", "Türkiye"))))))))))))))))))))))))
    AS ?countryName).

  # Measure Node URI
  BIND (URI(CONCAT("http://example.org/measure/", STR(?countryCodeA2), "/", STR(?icaoID), "/", STR(?TIME_PERIOD), "/",
    STR(?tra_meas), "/", STR(?tra_cov), "/", STR(?airline))) AS ?measure).

  # Airport Node URI
  BIND (URI(CONCAT("http://example.org/airport/", STR(?countryCodeA2), "/", STR(?icaoID))) AS ?airport).
  FILTER(?freq="A") # Just get annual data is enough
}
```

Structure



□ Nombre de triples : 11 536 926

Exemple de requêtes

Voici trois exemples de requêtes sur notre dataset:

- ❑ Liste des 20 aéroports français ayant déclaré le plus de vols en 2020.
- ❑ Les aéroports français dont les vols nationaux ont été les plus chargés en fonction des années
- ❑ Aéroports offrant un nombre de sièges passagers supérieurs à 100 000 en 2020.

Liste des 20 aéroports français ayant déclaré le plus de vols en 2020

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX ex: <http://example.org/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

SELECT ?icaoCode ?value
WHERE {
  ?x ex:unit "FLIGHT";      #Unité de mesure en nombre de vol
    ex:airline "TOTAL";     #Toutes les compagnies
    ex:coverageTransport "TOTAL"; #Tous les types de transports
    ex:measuredMetric "CAF";  #Tous les vols commerciaux (passagers, fret et courrier)
    dbo:year 2020;           #Vols ayant eu lieu en 2020
    rdf:value ?value;        #Récupération de la valeur du nombre de vol
    ex:declarativeAirport ?airport.

  ?airport dbo:iso31661Code "FR";           #Aéroports français
            dbo:icaoLocationIdentifier ?icaoCode. #Récupération du code ICAO de l'aéroport
}
ORDER BY DESC(?value) #Tri par ordre décroissant de nombre de vols
LIMIT 20              #Limitation à 20 résultats
```

La requête

Liste des 20 aéroports français ayant déclaré le plus de vols en 2020

icaoCode		value
LFPG	Paris-Charles de Gaulle	205924
LFPO	Paris-Orly	82865
LFMN	Nice Côte d'Azur	58725
LFLL	Lyon-Saint Exupéry	42546
LFML	Marseille Provence	41992
LFBO	Toulouse-Blagnac	35271
LFRS	Nantes Atlantique	23102
LFBD	Bordeaux Merignac	22760
TFFR	Pointe-à-Pitre Le Raizet (Guadeloupe)	15421
LFKJ	Ajaccio Napoléon Bonaparte	9829
LFOB	Paris Beauvais	9667
LFKB	Bastia Poretta	9626
LFMT	Montpellier Méditerranée	9341

Les premières lignes du résultat renvoyé par la requête

Les aéroports français dont les vols nationaux ont été les plus chargés en fonction des années

```
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
```

```
PREFIX ex: <http://example.org/>
```

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
```

```
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
# Affichage des données intéressantes ainsi que le pourcentage d'occupation
```

```
SELECT ?icao ?transportedPass ?nbOfSeats ((xsd:float(xsd:integer(?transportedPass))/xsd:integer(?nbOfSeats)) * 100 AS ?usageRate) ?year
```

```
WHERE {
```

```
  # Récupération le nombre de passagers transportées
```

```
  ?x    dbo:year ?year;
        ex:declarativeAirport ?airport;
        ex:measuredMetric "PAS_CRD";
        ex:coverageTransport "NAT";
        ex:airline "TOTAL";
        rdf:value ?transportedPass.
```

```
  # Récupération le nombre de places offertes
```

```
  ?y    dbo:year ?year;
        ex:declarativeAirport ?airport;
        ex:measuredMetric "ST_PAS";
        ex:coverageTransport "NAT";
        ex:airline "TOTAL";
        rdf:value ?nbOfSeats.
```

```
  # Récupération des détails de l'aéroport
```

```
  ?airport    dbo:country "France";
               dbo:icaoLocationIdentifier ?icao.
```

```
  # On évite la division par zéro impossible
```

```
  # Et on filtre, les valeurs qui donneraient des résultats éronnés
```

```
  FILTER(?nbOfSeats > 0 && ?transportedPass <= ?nbOfSeats)
```

```
}
```

```
ORDER BY DESC(?usageRate)
```

```
LIMIT 30
```

La requête

Les aéroports français dont les vols nationaux ont été les plus chargés en fonction des années

icao		transportedPass	nbOfSeats	usageRate	year
LFPG	Paris-Charles de Gaulle	5147608	5314808	96.85407	2009
LFPO	Paris-Orly	13760062	14749548	93.29141	2009
FMCZ	Dzaoudzi-Pamandzi (Mayotte)	268187	294240	91.14567	2019
LFMU	Béziers Cap d'Agde	50566	57469	87.988304	2017
LFOT	Tours - Val de Loire	25493	29081	87.66205	2015
LFMU	Béziers Cap d'Agde	59198	67788	87.32814	2019
LFOB	Paris Beauvais	87329	100059	87.27751	2019
FMEE	La Réunion Roland Garros	1312648	1505486	87.19098	2014
LFOT	Tours - Val de Loire	24721	28389	87.0795	2016
FMEE	La Réunion Roland Garros	1357728	1560684	86.995705	2015
FMEE	La Réunion Roland Garros	1326499	1530010	86.698715	2013
LFMU	Béziers Cap d'Agde	60118	69435	86.581696	2018
LFOB	Paris Beauvais	91624	105933	86.49241	2018
FMEE	La Réunion Roland Garros	1628214	1897342	85.81552	2019

Les premières lignes du résultat renvoyé par la requête

Aéroports offrant un nombre de sièges passagers supérieurs à 100 000 en 2020

PREFIX dbo: <http://dbpedia.org/ontology/>

PREFIX ex: <http://example.org/>

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

Affichage de nombre des sièges passagers offerts qui est supérieur à 100 000
par ordre croissant en 2020 et l'affichage est limité à 40 entités

```
SELECT ?countryName ?icao ?nbOfSeat
WHERE {
  # Récupération de nombre de sièges passagers offerts.
  ?x dbo:year 2020;
     ex:declarativeAirport ?airport;
     ex:measuredMetric "ST_PAS";
     ex:coverageTransport "TOTAL";
     ex:airline "TOTAL";
     rdf:value ?nbOfSeat.
  # Récupération de code de pays et de l'identifiant des aéroports
  ?airport dbo:country ?countryName;
           dbo:icaoLocationIdentifier ?icao.
  FILTER(?nbOfSeat>=100000)
}

ORDER BY ASC (?nbOfSeat)
LIMIT 40
```

La requête

Aéroports offrant un nombre de sièges passagers supérieurs à 100 000 en 2020

countryName	icao		nbOfSeat
Finland	EFKU	Kuopio	105070
Greece	LGKV	Kavala - Alexendre-le-Grand	106865
Norway	ENST	Sandnessjøen	107500
Poland	EPLL	Władysław Reymont de Łódź	108655
Portugal	LPAZ	Santa Maria	110006
France	LFMU	Béziers Cap d'Agde	110373
Finland	EFVA	Vaasa	110987
Sweden	ESMQ	Kalmar (Kalmar Öland)	111872
Sweden	ESNN	Sundsvall-Timrå	112911
Sweden	ESDF	Ronneby	114065
Austria	LOWL	Linz	115320
Türkiye	LTCM	Sinop	115331
Finland	EFKS	Kuusamo	115382
Iceland	BIAR	Akureyri	116404
Norway	ENOV	Ørsta-Volda	116457

Les premières lignes du résultat renvoyé par la requête

Liaison de notre dataset à d'autres datasets

Étape n°2



Liaisons possibles

En parcourant l'ensemble des datasets des groupes, nous avons déterminé les liaisons possibles avec ces derniers.

- ☐ **Groupe n°2** : Review de produit Amazon
- ☐ **Groupe n°3** : Gapminder Dataset
- ☐ **Groupe n°4** : CO2 Emissions Around the World
- ☐ **Groupe n°7** : Les levées de fonds des entreprises
- ☐ **Groupe n°9** : La déforestation par pays entre 1990 et 2015
- ☐ **Groupe n°11** : Part de chaque méthode de production électrique par pays


La principale difficulté

Pour pouvoir lier des datasets, il faut avoir des attributs en commun et surtout de même type dans notre cas, c'est ce point-là qui va poser problème.

Notamment la manière de représenter un pays.

- ❑ **Par nom de pays** : Incompatible avec notre dataset puisque pour les pays sont représentés par des codes
- ❑ **Par code pays** : Possible mais parfois incompatible, nous utilisons la norme ISO3166-1 alpha-2 (FR, US...) or certains groupes utilisent la norme ISO3166-1 alpha-3 (FRA, USA...)

La solution employée



- Ajout d'un attribut "*dbo:iso31661Code*" pour le code Alpha3
- Ajout d'un attribut "*dbo:country*" pour le nom du pays

Liaisons réalisées

- ☐ **Groupe n°3** : Le nombre de vols de passagers est-il relatif à l'HDI d'un pays en 2017 ?
- ☐ **Groupe n°4** : Affichage de la quantité de Co2 émise et du nombre de vols par pays en 2010
- ☐ **Groupe n°7** : Le nombre de vols est-il relatif au total des levées de fonds par pays en 2014 ?

Le nombre de vols de passagers est-il relatif à l'HDI d'un pays en 2017 ?

PREFIX ex: <http://example.org/>

PREFIX dbp: <http://dbpedia.org/property/>

PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

PREFIX dbo: <http://dbpedia.org/ontology/>

```
#
#  REQUÊTE AVEC LE GROUPE 3      (Fichier DataWithTarql.ttl pris le 04.12.2022)
#  Affiche le nom du pays, l'HDI du pays mais également le nombre de vols
#  commerciaux de passagers effectués par les principaux aéroports de ce
#  dernier en 2017
#
```

```
SELECT ?countryName ?hdi (SUM(?nbVolsPassagers) AS ?TotalVolsPassagers)
```

```
WHERE {
  ?x ex:declarativeAirport/dbo:country ?countryName;
     dbo:year 2017; # On spécifie l'année
     ex:unit "FLIGHT"; # On veut le nombre de vols
     ex:airline "TOTAL"; # Toutes compagnies confondues
     ex:coverageTransport "TOTAL"; # Toutes destinations confondues
     ex:measuredMetric "CAF-PAS"; # Uniquement les vols commerciaux de passagers
     rdf:value ?nbVolsPassagers.
}
```

```
SERVICE <http://127.0.0.1:3030/groupe3/query>{
  ?y dbp:country ?countryName;
     dbp:year "2017"^^xsd:int; # Tricky thing to make it work...
     dbp:humanDevelopmentIndex ?hdi.
}
GROUP BY ?countryName ?hdi
ORDER BY DESC (?TotalVolsPassagers)
```

La requête

Le nombre de vols de passagers est-il relatif à l'HDI d'un pays en 2017 ?

countryName	hdi	TotalVolsPassagers
United Kingdom	0.926e0	2188031
Germany	0.943e0	1908084
Spain	0.903e0	1761885
France	0.897e0	1508478
Italy	0.886e0	1321816
Norway	0.954e0	624643
Netherlands	0.939e0	539831
Greece	0.879e0	475309
Sweden	0.942e0	457463
Switzerland	0.949e0	453459
Portugal	0.858e0	405127
Poland	0.873e0	343662
Denmark	0.936e0	328205
Belgium	0.929e0	280487
Austria	0.919e0	271449
Ireland	0.947e0	248465
Finland	0.935e0	223733

Les premières lignes du résultat renvoyé par la requête

Affichage de la quantité de Co2 et du nombre de vols (passagers, frets et courrier) par pays en 2010

```
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX ex: <http://example.org/>
PREFIX ex2: <http://ex.org/#>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbpedia: <http://dbpedia.org/ontology>
```

```
#
#  REQUÊTE AVEC LE GROUPE 4      (Fichier CO2.V3.ttl pris le 04.12.2022)
#  Affiche le nom du pays, la quantité de Co2 émis par le pays en Tonne Cubique mais également
#  le nombre de vols (passagers, courrier et fret inclus) effectués par les principaux aéroports
#  de ce dernier en 2010
#  Le tout trié de manière décroissante sur le nombre de vols
#
```

```
SELECT ?countryName (xsd:float(?valeur) AS ?valeurTonneCubique) (SUM(?nbVols) AS ?TotalNbVols)
WHERE {
  ?measure ex:declarativeAirport ?airport;
            dbo:year 2010; # On spécifie l'année
            ex:unit "FLIGHT"; # On veut le nombre de vols
            ex:airline "TOTAL"; # Toutes compagnies confondues
            ex:coverageTransport "TOTAL"; # Toutes destinations confondues
            ex:measuredMetric "CAF"; # Passagers, courrier et fret
            rdf:value ?nbVols.

  ?airport dbo:iso31661Code ?countryCode;
            dbo:country ?countryName.

  SERVICE <http://127.0.0.1:3030/groupe4/sparql>{
    ?x dbo:iso31661Code ?countryCode;
        ex2:data ?data.

    ?data dbpedia:year 2010; # On spécifie l'année
           ex2:value ?valeur.
  }
}
```

```
GROUP BY ?countryName ?valeur
ORDER BY DESC (?TotalNbVols)
```

La requête

Affichage de la quantité de Co2 et du nombre de vols (passagers, frets et courrier) par pays en 2010

countryName	valeurTonneCubique	TotalNbVols
Germany	9.453399722e0	2022180
United Kingdom	7.686282334e0	1972059
Spain	5.866642426e0	1742531
France	5.348198409e0	1523827
Italy	6.836836177e0	1374719
Norway	8.469598181e0	554156
Netherlands	10.29828126e0	431312
Greece	7.874949777e0	422515
Sweden	5.132155515e0	414293
Switzerland	5.777702857e0	401461
Denmark	8.673891232e0	331233
Austria	8.365014901e0	315277
Belgium	9.794792242e0	305047
Portugal	4.817886772e0	283838
Poland	8.247028077e0	254653
Finland	11.65875348e0	250371
Ireland	8.813296615e0	209316

Les premières lignes du résultat renvoyé par la requête

Le nombre de vols est-il relatif au total des levées de fonds par pays en 2014 ?

```
PREFIX schema: <http://schema.org/>
PREFIX exA: <http://example.org/>
PREFIX ex: <http://example.com/>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
#
#  REQUÊTE AVEC LE GROUPE 7  (Fichier companies_laison.ttl pris le 04.12.2022)
#  Affiche le nom du pays, le total des valeurs de levées de fond du pays mais
#  également le nombre de vols commerciaux (passagers, fret et courrier)
#  effectués par les principaux aéroports de ce dernier en 2014
#
```

```
SELECT DISTINCT ?countryName ((xsd:integer(SUM(xsd:float(?amount)))) AS ?TotalAmount) (SUM(?nbVols) AS ?TotalNbVols)
WHERE {
    ?airport      dbo:iso31661Code      ?countryCode;
                  dbo:country ?countryName.

    ?measure      exA:declarativeAirport ?airport;
                  dbo:year               2014; # On spécifie l'année
                  exA:unit                "FLIGHT"; # On veut le nombre de vols
                  exA:airline             "TOTAL"; # Toutes compagnies confondues
                  exA:coverageTransport  "TOTAL"; # Toutes destinations confondues
                  exA:measuredMetric      "CAF"; # Passagers, courrier et fret
                  rdf:value                ?nbVols.

    SERVICE <http://127.0.0.1:3030/groupe7/query>{
        ?y      dbo:iso31661Code      ?countryCode;
                 ex:year               "2014.0"; # Pretty ugly but needed to make it work
                 schema:amount         ?amount.
    }
}
GROUP BY ?countryName
ORDER BY DESC (?TotalAmount)
```


Le nombre de vols est-il relatif au total des levées de fonds par pays en 2014 ?

countryName	TotalAmount	TotalNbVols
Germany	1327739392	61734720
United Kingdom	1079637760	108281120
France	96296008	7588600
Portugal	88000000	617090
Italy	77571056	9793248
Finland	53768828	1844192
Netherlands	52667584	5971728
Spain	28333152	17132599
Sweden	26177200	2694210
Ireland	24621664	3039750
Türkiye	17850000	4639348
Denmark	16920150	1985844
Poland	16560000	1182180
Norway	15966000	1871502
Switzerland	12449752	1772588
Austria	8048508	1148308

Les premières lignes du résultat renvoyé par la requête

Inférences

Étape n°3




Liaison de notre dataset au cloud LinkedData

Étape n°4



Comment se lier avec le cloud LinkedData ?

Utilisation d'attributs communs et universels pouvant servir de liens tel que:



Description avec le vocabulaire VOID

Étape n°5



Notre vocabulaire VOID

```
@prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#> .
@prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#> .
@prefix foaf: <http://xmlns.com/foaf/0.1/> .
@prefix dcterms: <http://purl.org/dc/terms/> .
@prefix void: <http://rdfs.org/ns/void#> .
```

```
:DBpedia rdf:type void:Dataset ;
  dcterms:title "Airline traffic data by main European airport" ;
  dcterms:description "This dataset provides us with data on the number of passengers, flights, seats or cargo weight
for European airports from 1993 to 2022 (still updated). Other filters exist for more details, please consult the web
page of the data producer." ;
  dcterms:source <https://ec.europa.eu/eurostat/databrowser/view/AVIA_TF_ALA/default/table> ;
  dcterms:publisher :data_europa;
  dcterms:contributor :Quentin_GOMES_DOS_REIS ;
  dcterms:contributor :Alan_PROVOST ;
  dcterms:contributor :Izzedine_issa_AHMAT ;
  dcterms:modified "2022-10-24"^^xsd:date ;

void:triples 11536926;
void:classes 2;
void:properties 9;

void:classPartition [ void:class dbo:Airport; ];
void:classPartition [ void:class ex:AirportMetricValue; ];

void:propertyPartition [ void:property ex:declarativeAirport; ];
void:propertyPartition [ void:property dbo:year; ];
void:propertyPartition [ void:property rdf:value; ];
void:propertyPartition [ void:property ex:unit; ];
void:propertyPartition [ void:property ex:measuredMetric; ];
void:propertyPartition [ void:property ex:coverageTransport; ];
void:propertyPartition [ void:property ex:airline; ];
```

```
void:propertyPartition [ void:property dbo:iso31661Code; ];
void:propertyPartition [ void:property dbo:icaoLocationIdentifier; ].

:data_europa rdf:type foaf:Organization;
  rdfs:label "ec.europa.eu" ;
  foaf:homepage <https://ec.europa.eu/eurostat> .

:Quentin_GOMES_DOS_REIS rdf:type foaf:Person ;
  rdfs:label "Quentin GOMES DOS REIS" ;
  foaf:mbox <quentin.gomes-dos-reis@etu.univ-nantes.fr>.

:Alan_PROVOST rdf:type foaf:Person ;
  rdfs:label "Alan PROVOST" ;
  foaf:mbox <alan.provost@etu.univ-nantes.fr>.

:Izzedine_issa_AHMAT rdf:type foaf:Person ;
  rdfs:label "Izzedine Issa AHMAT" ;
  foaf:mbox <izzedine-issa.ahmat@etu.univ-nantes.fr>.
```